************************************************************************************************************
*Replication*
*Do Gains in Political Representation Sweeten Tax Reform in China? It Depends on Who You Ask*


*Political Science Research and Methods*

*This version: October 11, 2022*
*Stata 17*
************************************************************************************************************

************************
*Preface*
************************

*ssc install coefplot, replace
*ssc install estout, replace



*Change this directory to your local folder
cd "/Users/JayCK/Library/CloudStorage/Box-Box/VAT_Project_New/Replication"

log using "log_file.txt", replace text

clear
estimates clear 
set more off

************************************************************************
*Figures and tables in the main text*
************************************************************************



use "china_ntwr.dta", clear


global baseline "i.representation i.pub_goods_services i.tax i.rate"
global interaction "elite_t10##i.representation elite_t10##i.pub_goods_services elite_t10##i.tax elite_t10##i.rate"
global coefinter1 "1.elite_t10#1.tax"  
global coefinter2 "1.elite_t10#1.rate 1.elite_t10#2.rate 1.elite_t10#3.rate 1.elite_t10#4.rate" 
global coefinter3 "1.elite_t10#1.pub_goods_services 1.elite_t10#2.pub_goods_services 1.elite_t10#3.pub_goods_services 1.elite_t10#4.pub_goods_services" 
global coefinter4 "1.elite_t10#1.representation 1.elite_t10#2.representation 1.elite_t10#3.representation 1.elite_t10#4.representation"


***********************************************
*FIGURE 2. China Conjoint Experiment by Elite Status*
***********************************************

reg selected_2 $baseline if elite_t10==1, cluster(psid)
estimates store R3_china_elite
reg selected_2 $baseline if elite_t10==0, cluster(psid)
estimates store R3_china_non_elite

coefplot /// 
	(R3_china_elite, label(Business Elites) msymbol(o) mcolor(black) ciopts(lc(black) lwidth(thin) lpattern(solid))) ///
	(R3_china_non_elite, label(Nonelites) msymbol(oh) mcolor(gray) ciopts(lc(gray) lwidth(thin) lpattern(shortdash))) ///
	, legend(row(1)) coeflabel(, wrap(25) labs(2)) omitted base ///
	xline(0) xlabel(-0.4(0.1)0.4) ylabel(, labs(small)) xtitle("Change in Pr(Tax Reform Policy selected)", size(small)) ///
	ciopts(width(thin)) xsize(11) ysize(9) scale(.9) graphregion(fcolor(white)) ///
	keep(*.tax *.rate *.pub_goods_services *.representation) ///
	headings(0.tax = "{bf:Tax Type}" ///
			 0.rate = "{bf:Tax Rate}" ///
			 0.pub_goods_services = "{bf:Government Services}" ///
			 0.representation = "{bf:Institutionalized Pol. Infl.}", labs(small)) 
			 
graph export "Figure 2.pdf", replace


***************************************************************************
*FIGURE 3. Differences in Conjoint Estimates Between Business Elites and Nonelites*
***************************************************************************

reg selected_2 $interaction, cluster(psid)

coefplot ///
	, xline(0) xlabel(-0.3(0.1)0.3) ylabel(, labs(small)) xtitle("Change in Pr(Tax Reform Policy selected)",size(small)) ///
	coeflabels(, wrap(43) labs(2) interaction(" x ")) ///
	msymbol(o) mcolor(black) ciopts(lc(black) lwidth(thin) lpattern(solid)) graphregion(fcolor(white)) ///
	keep($coefinter1 $coefinter2 $coefinter3 $coefinter4) ///
	headings(1.elite_t10#1.tax = "{bf:Tax Type}" ///
			 1.elite_t10#1.rate = "{bf:Tax Rate}" ///
			 1.elite_t10#1.pub_goods_services = "{bf:Government Services}" ///
			 1.elite_t10#1.representation = "{bf:Institutionalized Pol. Infl.}", labs(small))

graph export "Figure 3.pdf", replace


*******************************************************
*FIGURE 4. Direct Request about Preferences by Elite Status*
*******************************************************

reg tax_bargain i.elite_t10 if profile == 1 & task == 1
margins elite_t10
marginsplot ///
	, recast(scatter) scheme(s1mono) xscale(range(-.5 1.5) titlegap(1)) graphregion(color(white)) ///
	ytitle("Institutionalized Political Influence " "Preferred Over Government Service", size(medlarge)) title("") ///
	xlabel(0 "Nonelites" 1 "Business Elites", labels labsize(large)) xtitle("") ylabel(0(0.2)0.8)

graph export "Figure 4.pdf", replace


*************************************************************
*FIGURE 5. Trust in Government and Preference for IPI among Business Elites*
*************************************************************

reg tax_bargain i.trust_gov if profile == 1 & task == 1 & elite_t10 == 1, r
margins trust_gov
marginsplot ///
	, recast(scatter) scheme(s1mono) xscale(range(-.5 1.5) titlegap(1)) graphregion(color(white)) ///
	ytitle("Institutionalized Political Influence" "Preferred Over Government Service", size(medlarge)) title("")  ///
	xlabel(0 "No Trust" 1 "Trust", labels labsize(medlarge)) xtitle("Trust in Government") ylabel(0(0.2)0.9)

graph export "Figure 5.pdf", replace


***********************************************************************************************
*FIGURE 6. Conjoint Estimates of Elites and Nonelites for Tax Reforms Involving an Income Tax and VAT, separately*
***********************************************************************************************

*Income tax (6-a)

reg selected_2 i.representation i.pub_goods_services i.rate  if elite_t10==1 & tax==0,cluster(psid)
estimates store income_china_elite
reg selected_2 i.representation i.pub_goods_services i.rate  if elite_t10==0 & tax==0,cluster(psid)
estimates store income_china_nonelite

coefplot ///
	(income_china_elite, label(Business Elites) msymbol(o) mcolor(black) ciopts(lc(black) lwidth(thin) lpattern(solid))) ///
	(income_china_nonelite, label(Nonelites) msymbol(oh) mcolor(gray) ciopts(lc(gray) lwidth(thin) lpattern(shortdash))) ///
	, legend(row(1)) coeflabel(, wrap(25) labs(2)) omitted base ///
	xline(0) xlabel(-0.4(0.1)0.4) ylabel(, labs(medsmall)) xtitle("Change in Pr(Tax Reform Policy selected)", size(small)) ///
	ciopts(width(thin)) xsize(11) ysize(9) scale(.9) graphregion(fcolor(white)) ///
	keep(*.tax *.rate *.pub_goods_services *.representation) ///
	headings(0.tax = "{bf:Tax Type}" ///
			 0.rate = "{bf:Tax Rate}" ///
			 0.pub_goods_services = "{bf:Government Services}"  ///
			 0.representation = "{bf:Institutionalized Pol. Infl.}", labs(medsmall))
	
graph export "Figure 6(a).pdf", replace

	
*VAT (6-b)

reg selected_2 i.representation i.pub_goods_services i.rate  if elite_t10==1 & tax==1,cluster(psid)
estimates store vat_china_elite
reg selected_2 i.representation i.pub_goods_services i.rate  if elite_t10==0 & tax==1,cluster(psid)
estimates store vat_china_nonelite

coefplot ///
	(vat_china_elite, label(Business Elites) msymbol(o) mcolor(black) ciopts(lc(black) lwidth(thin) lpattern(solid))) ///
	(vat_china_nonelite, label(Nonelites) msymbol(oh) mcolor(gray) ciopts(lc(gray) lwidth(thin) lpattern(shortdash))) ///
	, legend(row(1)) coeflabel(, wrap(25) labs(2)) omitted base ///
	xline(0) xlabel(-0.4(0.1)0.4) ylabel(, labs(medsmall)) xtitle("Change in Pr(Tax Reform Policy selected)", size(small)) ///
	ciopts(width(thin)) xsize(11) ysize(9) scale(.9) graphregion(fcolor(white)) ///
	keep(*.tax *.rate *.pub_goods_services *.representation) ///
	headings(0.tax = "{bf:Tax Type}" ///
			 0.rate = "{bf:Tax Rate}" ///
			 0.pub_goods_services = "{bf:Government Services}" ///
			 0.representation = "{bf:Institutionalized Pol. Infl.}", labs(medsmall))

graph export "Figure 6(b).pdf", replace


***********************************************************************************************
*FIGURE 7. Difference in Conjoint Estimates between Elites and Nonelites for Tax Reforms w/ Low and High Tax Rates*
***********************************************************************************************

* Low rate (1, 5, 10%)

reg selected_2 elite_t10##i.representation elite_t10##i.pub_goods_services elite_t10##i.tax if rate ==0 | rate==1 | rate==2, cluster(psid)
estimates store R3_China_low_rate


coefplot R3_China_low_rate ///
	, xline(0) xlabel(-0.3(0.1)0.3) ylabel(, labs(medsmall)) xtitle("Change in Pr(Tax Reform Policy selected)",size(small)) ///
	coeflabels(, wrap(43) labs(2) interaction(" x ")) ///
	msymbol(o) mcolor(black) ciopts(lc(black) lwidth(thin) lpattern(solid)) graphregion(fcolor(white)) ///
	keep($coefinter1 $coefinter2 $coefinter3 $coefinter4) ///
	headings(1.elite_t10#1.tax = "{bf:Tax Type}" ///
			 1.elite_t10#1.rate = "{bf:Tax Rate}" ///
			 1.elite_t10#1.pub_goods_services = "{bf:Government Services}" ///
			 1.elite_t10#1.representation = "{bf:Institutionalized Pol. Infl.}", labs(medsmall))

graph export "Figure 7(a).pdf", replace		 

			 
* High rate (15% and 20%)

reg selected_2 elite_t10##i.representation elite_t10##i.pub_goods_services elite_t10##i.tax if rate ==3 | rate==4, cluster(psid)
estimates store R3_China_high_rate

coefplot R3_China_high_rate ///
	, xline(0) xlabel(-0.3(0.1)0.3) ylabel(, labs(medsmall)) xtitle("Change in Pr(Tax Reform Policy selected)",size(small)) ///
	coeflabels(, wrap(43) labs(2) interaction(" x ")) ///
	msymbol(o) mcolor(black) ciopts(lc(black) lwidth(thin) lpattern(solid)) graphregion(fcolor(white)) ///
	keep($coefinter1 $coefinter2 $coefinter3 $coefinter4) ///
	headings(1.elite_t10#1.tax = "{bf:Tax Type}" ///
			 1.elite_t10#1.rate = "{bf:Tax Rate}" ///
			 1.elite_t10#1.pub_goods_services = "{bf:Government Services}" ///
			 1.elite_t10#1.representation = "{bf:Institutionalized Pol. Infl.}", labs(medsmall))

graph export "Figure 7(b).pdf", replace


* Online Appendix 
													
***********************************************
*TABLE A-1. Summary statistics for Business Elite Sample*
***********************************************

global descriptive "selected_2 tax_bargain trust_gov money_today year_paying_tax awareness_vat_included edu_sat_dummy health_sat_dummy infr_sat_dummy environ_sat_dummy male age married ccp income_cat edu"

* Original elite sample
estpost sum $descriptive if profile==1 & task==1 & elite_1==1
esttab using "Table A-1_original_elite.csv", replace cell((mean sd min max count)) nonumber noobs title("Original Elite Sample")

* Screened elite sample
estpost sum $descriptive if profile==1 & task==1 & elite_t10==1
esttab using "Table A-1_screened_elite.csv", replace cell((mean sd min max count)) nonumber nomtitle noobs title("Screened Elite Sample")


	
***********************************************
*TABLE A-2. Summary statistics for Nonelite Sample*
***********************************************

* Original nonelite sample
estpost sum $descriptive if profile==1 & task==1 & elite_1==0
esttab using "Table A-2_original_nonelite.csv", replace cell((mean sd min max count)) nonumber noobs title("Original Nonelite Sample")

* Screened nonelite sample
estpost sum $descriptive if profile==1 & task==1 & elite_t10==0
esttab using "Table A-2_screened_nonelite.csv", replace cell((mean sd min max count)) nonumber noobs title("Screened Nonelite Sample")


***********************************************
*TABLE A-4. Main Conjoint Experiment by Elite Status*
***********************************************

global controls "male age i.edu i.sector ccp married i.income_cat"

reg selected_2 $baseline  if elite_t10==1,cluster(psid)
outreg2 using "Table A-4.tex", replace dec(3) label noomitted nobase nocons addtext(Socio-Economic Controls, No) ctitle("Business Elites")

reg selected_2 $baseline $controls  if elite_t10==1,cluster(psid)
outreg2 using "Table A-4.tex", append dec(3) label noomitted nobase nocons addtext(Socio-Economic Controls, Yes) drop($controls) ctitle("Business Elites")

reg selected_2 $baseline  if elite_t10==0,cluster(psid)
outreg2 using "Table A-4.tex", append dec(3) label noomitted nobase nocons addtext(Socio-Economic Controls, No) ctitle("Nonelites")

reg selected_2 $baseline $controls  if elite_t10==0,cluster(psid)
outreg2 using "Table A-4.tex", append dec(3) label noomitted nobase nocons addtext(Socio-Economic Controls, Yes) drop($controls) ctitle("Nonelites")


**********************************************************
*TABLE A-5. Main Conjoint Experiment with Elite Status Interaction*
**********************************************************

reg selected_2 $interaction ,cluster(psid)
outreg2 using "Table A-5.tex", replace dec(3) label noomitted nobase nocons addtext(Socio-Economic Controls, No) ///
	keep(i.elite_t10 i.tax i.rate i.pub_goods_services i.representation $coefinter1 $coefinter2 $coefinter3 $coefinter4 )
reg selected_2 $interaction $controls, cluster(psid)
outreg2 using "Table A-5.tex", append dec(3) label noomitted nobase nocons addtext(Socio-Economic Controls, Yes) ///
	keep(i.elite_t10 i.tax i.rate i.pub_goods_services i.representation $coefinter1 $coefinter2 $coefinter3 $coefinter4 )

	

****************************************************************************************************************************
*FIGURE A-3: Conjoint Experiment with Stricter Definition of Business Elite: Top 10%, 5%, and 1% firms, measured by total employees*
****************************************************************************************************************************

reg selected_2 $baseline  if elite_t10==1,cluster(psid)
estimates store R3_china_elite_top10
reg selected_2 $baseline  if elite_t5==1,cluster(psid)
estimates store R3_china_elite_top5
reg selected_2 $baseline  if elite_t1==1,cluster(psid)
estimates store R3_china_elite_top1

coefplot /// 
	(R3_china_elite_top10, label(Business Elites Top-10%) msymbol(o) mcolor(black) ciopts(lc(black) lwidth(thin) lpattern(solid))) ///
	(R3_china_elite_top5, label(Business Elites Top-5%) msymbol(d) mcolor(black) ciopts(lc(black) lwidth(thin) lpattern(solid))) ///
	(R3_china_elite_top1, label(Business Elites Top-1%) msymbol(t) mcolor(black) ciopts(lc(black) lwidth(thin) lpattern(solid))) ///
	, legend(row(1)) coeflabel(, wrap(22) labs(2)) omitted base ///
	xline(0) xlabel(-0.4(0.1)0.4) xtitle("Change in Pr(Tax Reform Policy selected)", size(small))  /// 
	xsize(11) ysize(20) scale(0.6) graphregion(fcolor(white)) ///
	keep(*.tax *.rate *.pub_goods_services *.representation) ///
	headings(0.tax = "{bf:Tax Type}" ///
			 0.rate = "{bf:Tax Rate}" ///
			 0.pub_goods_services = "{bf:Government Services}" ///
			 0.representation = "{bf:Institutionalized Pol. Infl.}")	

graph export "Figure A-3.pdf", replace

			 
***********************************************
*FIGURE A-4. Conjoint Analysis Using Unscreened Sample*
***********************************************

reg selected_2 $baseline  if elite_1==1,cluster(psid)
estimates store unscreened_elite
reg selected_2 $baseline  if elite_1==0,cluster(psid)
estimates store unscreened_nonelite

coefplot ///
	(unscreened_elite, label(Business Elites) msymbol(o) mcolor(black) ciopts(lc(black) lwidth(thin) lpattern(solid))) ///
	(unscreened_nonelite, label(Nonelites) msymbol(oh) mcolor(gray) ciopts(lc(gray) lwidth(thin) lpattern(shortdash))) ///
	, legend(row(1)) coeflabel(, wrap(22) labs(2)) omitted base ///
	xline(0) xlabel(-0.4(0.1)0.4) xtitle("Change in Pr(Tax Reform Policy selected)", size(small)) ///
	ciopts(width(thin)) xsize(11) ysize(9) scale(.9) graphregion(fcolor(white)) ///
	keep(*.tax *.rate *.pub_goods_services *.representation) ///
	headings(0.tax = "{bf:Tax Type}" ///
			 0.rate = "{bf:Tax Rate}" ///
			 0.pub_goods_services = "{bf:Government Services}" ///
			 0.representation = "{bf:Institutionalized Pol. Infl.}")	

graph export "Figure A-4.pdf", replace
		
		
**********************************************************
*FIGURE A-5. Conjoint Analysis with High-Income Ordinary Citizens*
**********************************************************


reg selected_2 $baseline  if elite_t10==. & income_cat>=8,cluster(psid)
estimates store R3_china_nonelite_rich

coefplot ///
	(R3_china_elite, label(Business Elites) msymbol(o) mcolor(black) ciopts(lc(black) lwidth(thin) lpattern(solid))) ///
	(R3_china_nonelite_rich, label(High-income Nonelites) msymbol(dh) mcolor(black) ciopts(lc(black) lwidth(thin) lpattern(solid))) ///
	(R3_china_non_elite, label(Nonelites) msymbol(oh) mcolor(black) ciopts(lc(black) lwidth(thin) lpattern(solid))) ///
	, legend(row(1)) coeflabel(, wrap(22) labs(2)) omitted base ///
	xline(0) xlabel(-0.4(0.1)0.4) xtitle("Change in Pr(Tax Reform Policy selected)", size(small)) ///
	ciopts(width(thin)) xsize(11) ysize(9) scale(.8) graphregion(fcolor(white)) ///
	keep(*.tax *.rate *.pub_goods_services *.representation) ///
	headings(0.tax = "{bf:Tax Type}" ///
			 0.rate = "{bf:Tax Rate}" ///
			 0.pub_goods_services = "{bf:Government Services}" ///
			 0.representation = "{bf:Institutionalized Pol. Infl.}")	

graph export "Figure A-5.pdf", replace


*******************************************
*FIGURE A-6. Conjoint Analysis by CCP membership*
*******************************************

reg selected_2 ccp##i.representation ccp##i.pub_goods_services ccp##i.tax ccp##i.rate if elite_t10==1 ,cluster(psid)
estimates store R3_China_cpp_diff

coefplot R3_China_cpp_diff ///
	, xline(0) xlabel(-0.3(0.1)0.3) xtitle("Change in Pr(Tax Reform Policy selected)",size(small)) ///
	coeflabels(, wrap(30) labs(2) interaction(" x "))  ///
	msymbol(o) mcolor(black) ciopts(lc(black) lwidth(thin) lpattern(solid)) graphregion(fcolor(white)) ///
	keep(1.ccp#1.tax ///
		 1.ccp#1.rate 1.ccp#2.rate 1.ccp#3.rate 1.ccp#4.rate ///
		 1.ccp#1.pub_goods_services 1.ccp#2.pub_goods_services 1.ccp#3.pub_goods_services 1.ccp#4.pub_goods_services ///
		 1.ccp#1.representation 1.ccp#2.representation 1.ccp#3.representation 1.ccp#4.representation) ///
	headings(1.ccp#1.tax = "{bf:Tax Type}" ///
			 1.ccp#1.rate = "{bf:Tax Rate}" ///
			 1.ccp#1.pub_goods_services = "{bf:Government Services}" ///
			 1.ccp#1.representation = "{bf:Institutionalized Pol. Infl.}")

graph export "Figure A-6.pdf", replace


***********************************************************************
*FIGURE A-7. CCP Membership, Trust, and Political Say Among Chinese Business Elites*
***********************************************************************

reg tax_bargain i.trust_gov if profile == 1 & task == 1 & elite_t10 == 1 & ccp==1, r level(90)
margins trust_gov
marginsplot ///
	, level(90) recast(scatter) scheme(s1mono) xscale(range(-.5 1.5) titlegap(1)) title("") ///
	xtitle("Trust in Government") xlabel(0 "No Trust" 1 "Trust", labels labsize(medlarge)) ///
	ytitle("Institutionalized Political Influence" "Preferred Over Government Service", size(medlarge)) ylabel(0(0.2)1)

graph export "Figure A-7.pdf", replace


************************************************
*FIGURE A-8. Conjoint Analysis Without CCP and SOE Members*
************************************************


reg selected_2 $baseline if elite_t10==1 & ccp!=1 ,cluster(psid)
estimates store R3_china_elite_noccp

reg selected_2 $baseline if elite_t10==1 & sector!=3,cluster(psid)
estimates store R3_china_elite_nosoe

coefplot ///
	(R3_china_elite, label(Business Elites) msymbol(oh) mcolor(black) ciopts(lc(black) lwidth(thin) lpattern(solid))) ///
	(R3_china_elite_noccp, label(Business Elites without CCP Members) msymbol(sh) mcolor(gray) ciopts(lc(gray) lwidth(thin) lpattern(shortdash))) ///
	(R3_china_elite_nosoe, label(Business Elites without SOE Members) msymbol(th) mcolor(navy) ciopts(lc(navy) lwidth(thin) lpattern(dash))) ///
	, legend(row(3)) coeflabel(, wrap(22) labs(2)) omitted base ///
	xline(0) xlabel(-0.4(0.1)0.4) xtitle("Change in Pr(Tax Reform Policy selected)", size(small)) ///
	ciopts(width(thin)) xsize(11) ysize(9) scale(.9) graphregion(fcolor(white)) ///
	keep(*.tax *.rate *.pub_goods_services *.representation) ///
	headings(0.tax = "{bf:Tax Type}" ///
			 0.rate = "{bf:Tax Rate}" ///
			 0.pub_goods_services = "{bf:Government Services}" ///
			 0.representation = "{bf:Institutionalized Pol. Infl.}")	

graph export "Figure A-8.pdf", replace


********************************************************************
*FIGURE A-9. ACME for Business Elites Working at Domestic and Foreign Firms*
********************************************************************

gen Foreign = .
replace Foreign =1 if sector ==6
replace Foreign =0 if sector ==5

reg selected_2 Foreign##i.representation Foreign##i.pub_goods_services Foreign##i.tax Foreign##i.rate if elite_t10==1,cluster(psid)

coefplot ///
	, xline(0) xlabel(-0.3(0.1)0.3) xtitle("Change in Pr(Tax Reform Policy selected)",size(small)) ///
	coeflabels(, wrap(30) labs(2) interaction(" x ")) ///
	msymbol(o) mcolor(black) ciopts(lc(black) lwidth(thin) lpattern(solid)) graphregion(fcolor(white)) ///
	keep(1.Foreign#1.tax ///
		 1.Foreign#1.rate 1.Foreign#2.rate 1.Foreign#3.rate 1.Foreign#4.rate ///
		 1.Foreign#1.pub_goods_services 1.Foreign#2.pub_goods_services 1.Foreign#3.pub_goods_services 1.Foreign#4.pub_goods_services ///
		 1.Foreign#1.representation 1.Foreign#2.representation 1.Foreign#3.representation 1.Foreign#4.representation ) ///
	headings(1.Foreign#1.tax = "{bf:Tax Type}" ///
			 1.Foreign#1.rate = "{bf:Tax Rate}" ///
			 1.Foreign#1.pub_goods_services = "{bf:Government Services}" ///
			 1.Foreign#1.representation = "{bf:Institutionalized Pol. Infl.}")

graph export "Figure A-9.pdf", replace			 


***************************
*FIGURE A-10. Social Desirability*
***************************

reg selected_2 $baseline if elite_t10==1 & trust_gov==0, cluster(psid)			
estimates store elite_notrust_china
 
reg selected_2 $baseline if elite_t10==0 & trust_gov==0, cluster(psid)			
estimates store nonelite_notrust_china

coefplot ///
	(elite_notrust_china, label(Business Elites) msymbol(o) mcolor(black) ciopts(lc(black) lwidth(thin) lpattern(solid))) ///
	(nonelite_notrust_china, label(Nonelites) msymbol(oh) mcolor(gray) ciopts(lc(gray) lwidth(thin) lpattern(shortdash))) ///
	, legend(row(1)) coeflabel(, wrap(22) labs(2)) omitted base ///
	xline(0) xlabel(-0.4(0.1)0.4) xtitle("Change in Pr(Tax Reform Policy selected)", size(small)) ///
	ciopts(width(thin)) xsize(11) ysize(9) scale(.9) graphregion(fcolor(white)) ///
	keep(*.tax *.rate *.pub_goods_services *.representation) ///
	headings(0.tax = "{bf:Tax Type}" ///
			 0.rate = "{bf:Tax Rate}" ///
			 0.pub_goods_services = "{bf:Government Services}" ///
			 0.representation = "{bf:Institutionalized Pol. Infl.}")	

graph export "Figure A-10.pdf", replace
			 

******************************************************************
*FIGURE A-11. Preference for IPI by VAT Awareness and Elite Status in China*
******************************************************************

* (a) VAT Awareness (Proportions)

reg awareness_vat_included_dmy i.elite_t10 if profile == 1 & task == 1, r
margins i.elite_t10
marginsplot ///
	, recast(scatter) scheme(s1mono) xscale(range(-.5 1.5) titlegap(1))  graphregion(color(white)) ///
	title("") ytitle(Group Average) ylabel(0(0.2)0.8) xtitle("") xlabel(0 "Nonelites" 1 "Business Elites", labels labsize(medium))  


graph export "Figure A-11(a).pdf", replace	


* (b) Differences between aware elites and aware nonelites

reg selected_2 $interaction if awareness_vat_included_dmy == 1 ,cluster(psid)
estimates store R3_china_elitenonelite_vat

coefplot R3_china_elitenonelite_vat ///
	, xline(0) xlabel(-0.3(0.1)0.3) xtitle("Change in Pr(Tax Reform Policy selected)",size(small)) ///
	coeflabels(, wrap(30) labs(2) interaction(" x ")) ///
	msymbol(o) mcolor(black) ciopts(lc(black) lwidth(thin) lpattern(solid)) graphregion(fcolor(white)) ///
	keep($coefinter1 $coefinter2 $coefinter3 $coefinter4) ///
	headings(1.elite_t10#1.tax = "{bf:Tax Type}" ///
			 1.elite_t10#1.rate = "{bf:Tax Rate}" ///
			 1.elite_t10#1.pub_goods_services = "{bf:Government Services}" ///
			 1.elite_t10#1.representation = "{bf:Institutionalized Pol. Infl.}")

graph export "Figure A-11(b).pdf", replace				 
			 
			 
* (c) Differences between aware and unaware nonelites

global interaction_vat "awareness_vat_included_dmy##i.representation awareness_vat_included_dmy##i.pub_goods_services awareness_vat_included_dmy##i.tax awareness_vat_included_dmy##i.rate"

reg selected_2 $interaction_vat if elite_t10==0, cluster(psid)
estimates store R3_china_nonelite_vat

coefplot R3_china_nonelite_vat ///
	, xline(0) xlabel(-0.3(0.1)0.3) xtitle("Change in Pr(Tax Reform Policy selected)",size(small)) ///
	coeflabels(, wrap(30) labs(2) interaction(" x ")) ///
	msymbol(o) mcolor(black) ciopts(lc(black) lwidth(thin) lpattern(solid)) graphregion(fcolor(white)) ///
	keep(1.awareness_vat_included_dmy#1.tax ///
		 1.awareness_vat_included_dmy#1.rate 1.awareness_vat_included_dmy#2.rate ///
		 1.awareness_vat_included_dmy#3.rate 1.awareness_vat_included_dmy#4.rate ///
		 1.awareness_vat_included_dmy#1.pub_goods_services 1.awareness_vat_included_dmy#2.pub_goods_services ///
		 1.awareness_vat_included_dmy#3.pub_goods_services 1.awareness_vat_included_dmy#4.pub_goods_services ///
		 1.awareness_vat_included_dmy#1.representation 1.awareness_vat_included_dmy#2.representation ///
		 1.awareness_vat_included_dmy#3.representation 1.awareness_vat_included_dmy#4.representation ) ///
	headings(1.awareness_vat_included_dmy#1.tax = "{bf:Tax Type}" ///
			 1.awareness_vat_included_dmy#1.rate = "{bf:Tax Rate}" ///
			 1.awareness_vat_included_dmy#1.pub_goods_services = "{bf:Government Services}" ///
			 1.awareness_vat_included_dmy#1.representation = "{bf:Institutionalized Pol. Infl.}")

graph export "Figure A-11(c).pdf", replace	


* (d) Differences between aware and unaware elites

reg selected_2 $interaction_vat if elite_t10==1, cluster(psid)
estimates store R3_china_elite_vat

coefplot R3_china_elite_vat ///
	, xline(0) xlabel(-0.3(0.1)0.3) xtitle("Change in Pr(Tax Reform Policy selected)",size(small)) ///
	coeflabels(, wrap(30) labs(2) interaction(" x ")) ///
	msymbol(o) mcolor(black) ciopts(lc(black) lwidth(thin) lpattern(solid)) graphregion(fcolor(white)) ///
	keep(1.awareness_vat_included_dmy#1.tax ///
		 1.awareness_vat_included_dmy#1.rate 1.awareness_vat_included_dmy#2.rate ///
		 1.awareness_vat_included_dmy#3.rate 1.awareness_vat_included_dmy#4.rate ///
		 1.awareness_vat_included_dmy#1.pub_goods_services 1.awareness_vat_included_dmy#2.pub_goods_services ///
		 1.awareness_vat_included_dmy#3.pub_goods_services 1.awareness_vat_included_dmy#4.pub_goods_services ///
		 1.awareness_vat_included_dmy#1.representation 1.awareness_vat_included_dmy#2.representation ///
		 1.awareness_vat_included_dmy#3.representation 1.awareness_vat_included_dmy#4.representation ) ///
	headings(1.awareness_vat_included_dmy#1.tax = "{bf:Tax Type}" ///
			 1.awareness_vat_included_dmy#1.rate = "{bf:Tax Rate}" ///
			 1.awareness_vat_included_dmy#1.pub_goods_services = "{bf:Government Services}" ///
			 1.awareness_vat_included_dmy#1.representation = "{bf:Institutionalized Pol. Infl.}")

graph export "Figure A-11(d).pdf", replace	
		

***************************************************
*TABLE A-6. How is VAT distributed between producers and consumers*
***************************************************
 
estpost tabulate vat_burden elite_t10 if profile==1 & task==1, nototal
esttab using "Table A-6.csv", replace cells("b(label(freq)) colpct(fmt(2)label(pct))")  unstack noobs nonumber nomtitle


******************************************************************
*FIGURE A-12. Conjoint Analysis by Time Horizon and Elite Status in China*
******************************************************************

* (a) Receive money today (proportion by elite status)
reg money_today i.elite_t10 if profile == 1 & task == 1, r
margins i.elite_t10
marginsplot ///
	, recast(scatter) scheme(s1mono) xscale(range(-.5 1.5) titlegap(1)) graphregion(color(white)) ///
	title("(A) Receive Money Today") ///
	xtitle("") xlabel(0 "Nonelites" 1 "Business Elites", labels labsize(medium)) ///
	ytitle(Group Average) ylabel(0(0.2)0.8)

graph export "Figure A-12(a).pdf", replace	

	
* (b) Elite-nonelite differences in conjoint estimates for respondents with long time horizons 
reg selected_2 $interaction if money_today == 0, cluster(psid)
coefplot ///
	, xline(0) xlabel(-0.3(0.1)0.3) xtitle("Change in Pr(Tax Reform Policy selected)",size(small)) ///
	coeflabels(, wrap(30) labs(2) interaction(" x ")) ///
	msymbol(o) mcolor(black) ciopts(lc(black) lwidth(thin) lpattern(solid)) graphregion(fcolor(white)) ///
	title("(B) Low Discounting Rate") ///
	keep($coefinter1 $coefinter2 $coefinter3 $coefinter4) ///
	headings(1.elite_t10#1.tax = "{bf:Tax Type}" ///
			 1.elite_t10#1.rate = "{bf:Tax Rate}" ///
			 1.elite_t10#1.pub_goods_services = "{bf:Government Services}" ///
			 1.elite_t10#1.representation = "{bf:Institutionalized Pol. Infl.}") 
			 
graph export "Figure A-12(b).pdf", replace	


* (c) Differences in conjoint estimates by time horizon among nonelites


reg selected_2 low_discount##i.representation low_discount##i.pub_goods_services ///
	low_discount##i.tax low_discount##i.rate if elite_t10==0, cluster(psid)

coefplot ///
	, xline(0) xlabel(-0.3(0.1)0.3) xtitle("Change in Pr(Tax Reform Policy selected)",size(small)) ///
	coeflabels(, wrap(30) labs(2) interaction(" x ")) ///
	msymbol(o) mcolor(black) ciopts(lc(black) lwidth(thin) lpattern(solid)) graphregion(fcolor(white)) ///
	title("(C) Nonelites") ///
	keep(1.low_discount#1.tax ///
		 1.low_discount#1.rate 1.low_discount#2.rate 1.low_discount#3.rate 1.low_discount#4.rate ///
		 1.low_discount#1.pub_goods_services 1.low_discount#2.pub_goods_services ///
		 1.low_discount#3.pub_goods_services 1.low_discount#4.pub_goods_services ///
		 1.low_discount#1.representation 1.low_discount#2.representation ///
		 1.low_discount#3.representation 1.low_discount#4.representation) ///
	headings(1.low_discount#1.tax = "{bf:Tax Type}" ///
			 1.low_discount#1.rate = "{bf:Tax Rate}" ///
			 1.low_discount#1.pub_goods_services = "{bf:Government Services}" ///
			 1.low_discount#1.representation = "{bf:Institutionalized Pol. Infl.}") 
			 
graph export "Figure A-12(c).pdf", replace	


**************************************************************************
*FIGURE A-13. Satisfaction with Government Services Goods by Elite Status in China*
**************************************************************************

*See NTWR R Script.R*


*********************************************************************************
*FIGURE A-14. Conjoint Analysis by Satisfaction with Public Goods for Chinese Business Elites*
*********************************************************************************

* (a) Education
label var edu_sat_dummy "Satisfied"

reg selected_2 i.edu_sat_dummy##i.representation i.edu_sat_dummy##i.pub_goods_services ///
	i.edu_sat_dummy##i.tax i.edu_sat_dummy##i.rate if elite_t10==1, cluster(psid)

coefplot ///
	, xline(0) xlabel(-0.3(0.1)0.3) xtitle("Change in Pr(Tax Reform Policy selected)",size(small)) ///
	coeflabels(, wrap(30) labs(2) interaction(" x ")) ///
	msymbol(o) mcolor(black) ciopts(lc(black) lwidth(thin) lpattern(solid)) graphregion(fcolor(white)) ///
	title((A) Education Satisfaction) ///
	keep(1.edu_sat_dummy#1.tax ///
		 1.edu_sat_dummy#1.rate 1.edu_sat_dummy#2.rate 1.edu_sat_dummy#3.rate 1.edu_sat_dummy#4.rate ///
		 1.edu_sat_dummy#1.pub_goods_services 1.edu_sat_dummy#2.pub_goods_services ///
		 1.edu_sat_dummy#3.pub_goods_services 1.edu_sat_dummy#4.pub_goods_services ///
		 1.edu_sat_dummy#1.representation 1.edu_sat_dummy#2.representation ///
		 1.edu_sat_dummy#3.representation 1.edu_sat_dummy#4.representation) ///
	headings(1.edu_sat_dummy#1.tax = "{bf:Tax Type}" ///
		     1.edu_sat_dummy#1.rate = "{bf:Tax Rate}" ///
		     1.edu_sat_dummy#1.pub_goods_services = "{bf:Government Services}" ///
		     1.edu_sat_dummy#1.representation = "{bf:Institutionalized Pol. Infl.}")

graph export "Figure A-14(a).pdf", replace	


* (b) Healthcare
label var health_sat_dummy "Satisfied"

reg selected_2 i.health_sat_dummy##i.representation i.health_sat_dummy##i.pub_goods_services ///
	i.health_sat_dummy##i.tax i.health_sat_dummy##i.rate if elite_t10==1, cluster(psid)

coefplot ///
	, xline(0) xlabel(-0.3(0.1)0.3) xtitle("Change in Pr(Tax Reform Policy selected)",size(small)) ///
	coeflabels(, wrap(30) labs(2) interaction(" x ")) ///
	msymbol(o) mcolor(black) ciopts(lc(black) lwidth(thin) lpattern(solid)) graphregion(fcolor(white)) ///
	title((B) Healthcare Satisfaction) ///
	keep(1.health_sat_dummy#1.tax ///
		 1.health_sat_dummy#1.rate 1.health_sat_dummy#2.rate 1.health_sat_dummy#3.rate 1.health_sat_dummy#4.rate ///
		 1.health_sat_dummy#1.pub_goods_services 1.health_sat_dummy#2.pub_goods_services ///
		 1.health_sat_dummy#3.pub_goods_services 1.health_sat_dummy#4.pub_goods_services ///
		 1.health_sat_dummy#1.representation 1.health_sat_dummy#2.representation ///
		 1.health_sat_dummy#3.representation 1.health_sat_dummy#4.representation) ///
	headings(1.health_sat_dummy#1.tax = "{bf:Tax Type}" ///
		     1.health_sat_dummy#1.rate = "{bf:Tax Rate}" ///
		     1.health_sat_dummy#1.pub_goods_services = "{bf:Government Services}" ///
		     1.health_sat_dummy#1.representation = "{bf:Institutionalized Pol. Infl.}")

graph export "Figure A-14(b).pdf", replace				 
 

* (c) Infrastructure
label var infr_sat_dummy "Satisfied"

reg selected_2 i.infr_sat_dummy##i.representation i.infr_sat_dummy##i.pub_goods_services ///
	i.infr_sat_dummy##i.tax i.infr_sat_dummy##i.rate if elite_t10==1, cluster(psid)

coefplot ///
	, xline(0) xlabel(-0.3(0.1)0.3) xtitle("Change in Pr(Tax Reform Policy selected)",size(small)) ///
	coeflabels(, wrap(30) labs(2) interaction(" x ")) ///
	msymbol(o) mcolor(black) ciopts(lc(black) lwidth(thin) lpattern(solid)) graphregion(fcolor(white)) ///
	title((C) Infrastructure Satisfaction) ///
	keep(1.infr_sat_dummy#1.tax ///
		 1.infr_sat_dummy#1.rate 1.infr_sat_dummy#2.rate 1.infr_sat_dummy#3.rate 1.infr_sat_dummy#4.rate ///
		 1.infr_sat_dummy#1.pub_goods_services 1.infr_sat_dummy#2.pub_goods_services ///
		 1.infr_sat_dummy#3.pub_goods_services 1.infr_sat_dummy#4.pub_goods_services ///
		 1.infr_sat_dummy#1.representation 1.infr_sat_dummy#2.representation ///
		 1.infr_sat_dummy#3.representation 1.infr_sat_dummy#4.representation) ///
	headings(1.infr_sat_dummy#1.tax = "{bf:Tax Type}" ///
		     1.infr_sat_dummy#1.rate = "{bf:Tax Rate}" ///
		     1.infr_sat_dummy#1.pub_goods_services = "{bf:Government Services}" ///
		     1.infr_sat_dummy#1.representation = "{bf:Institutionalized Pol. Infl.}") 
		 
graph export "Figure A-14(c).pdf", replace				 
			 
			 
* (d) Environment
label var environ_sat_dummy "Satisfied"

reg selected_2 i.environ_sat_dummy##i.representation i.environ_sat_dummy##i.pub_goods_services ///
	i.environ_sat_dummy##i.tax i.environ_sat_dummy##i.rate if elite_t10==1, cluster(psid)

coefplot ///
	, xline(0) xlabel(-0.3(0.1)0.3) xtitle("Change in Pr(Tax Reform Policy selected)",size(small)) ///
	coeflabels(, wrap(30) labs(2) interaction(" x ")) ///
	msymbol(o) mcolor(black) ciopts(lc(black) lwidth(thin) lpattern(solid)) graphregion(fcolor(white)) ///
	title((D) Environment Satisfaction) ///
	keep(1.environ_sat_dummy#1.tax ///
		 1.environ_sat_dummy#1.rate 1.environ_sat_dummy#2.rate 1.environ_sat_dummy#3.rate 1.environ_sat_dummy#4.rate ///
		 1.environ_sat_dummy#1.pub_goods_services 1.environ_sat_dummy#2.pub_goods_services ///
		 1.environ_sat_dummy#3.pub_goods_services 1.environ_sat_dummy#4.pub_goods_services ///
		 1.environ_sat_dummy#1.representation 1.environ_sat_dummy#2.representation ///
		 1.environ_sat_dummy#3.representation 1.environ_sat_dummy#4.representation) ///
	headings(1.environ_sat_dummy#1.tax = "{bf:Tax Type}" ///
		     1.environ_sat_dummy#1.rate = "{bf:Tax Rate}" ///
		     1.environ_sat_dummy#1.pub_goods_services = "{bf:Government Services}" ///
		     1.environ_sat_dummy#1.representation = "{bf:Institutionalized Pol. Infl.}") 
			 
graph export "Figure A-14(d).pdf", replace				 

			 
*************************************
*FIGURE A-17. Conjoint Analysis in Taiwan*
*************************************
use "taiwan_ntwr.dta", clear

reg selected_2 $baseline  if elite_t10==1, cluster(psid)
estimates store R3_taiwan_elite

reg selected_2 $baseline  if elite_t10==0,cluster(psid)
estimates store R3_taiwan_non_elite

coefplot ///
	(R3_taiwan_elite, label(Business Elites) msymbol(o) mcolor(black) ciopts(lc(black) lwidth(thin) lpattern(solid))) ///
	(R3_taiwan_non_elite, label(Nonelites) msymbol(oh) mcolor(gray) ciopts(lc(gray) lwidth(thin) lpattern(shortdash))) ///
	, legend(row(1)) coeflabel(, wrap(22) labs(2)) omitted base ///
	xline(0) xlabel(-0.4(0.1)0.4) xtitle("Change in Pr(Tax Reform Policy selected)", size(small)) ///
	ciopts(width(thin)) xsize(11) ysize(9) scale(.9) graphregion(fcolor(white)) ///
	keep(*.tax *.rate *.pub_goods_services *.representation) ///
	headings(0.tax = "{bf:Tax Type}" ///
			 0.rate = "{bf:Tax Rate}" ///
			 0.pub_goods_services = "{bf:Government Services}" ///
			 0.representation = "{bf:Institutionalized Pol. Infl.}") 

graph export "Figure A-17.pdf", replace				 


*The End*

clear
estimates clear 

log close
